Using MATLAB/Scilab, design a quarter-wave transformer to match a 40 Ω load to a 75 Ω transmission line at 1.5 GHz. Determine the transformer characteristic impedance and physical length.

Program code:

```
clc;
clear;
close;

// ------------------------------------------------
// QUARTER-WAVE TRANSFORMER DESIGN
// ------------------------------------------------

// Given parameters
Z0 = 75;            // Characteristic impedance of transmission line (ohms)
ZL = 40;            // Load impedance (ohms)
f0 = 1.5e9;         // Design frequency (Hz)
c = 3e8;            // Velocity of light (m/s)

// Transformer characteristic impedance
Zt = sqrt(Z0 * ZL);

// Wavelength at design frequency
lambda0 = c / f0;

// Quarter-wave transformer physical length
l = lambda0 / 4;
```

```
// Display calculated values
disp("==================================================");
disp("       QUARTER-WAVE TRANSFORMER");
disp("==================================================");

printf("Transmission line impedance = %.2f Ohm\n", Z0);
printf("Load impedance              = %.2f Ohm\n", ZL);
printf("Operating frequency         = %.2f GHz\n", f0/1e9);
printf("Wavelength                  = %.4f m\n", lambda0);
printf("Transformer impedance       = %.2f Ohm\n", Zt);
printf("Transformer length          = %.4f m\n", l);
printf("Transformer length          = %.2f cm\n", l*100);

// ------------------------------------------------------
// INPUT IMPEDANCE CALCULATION
// Zin = Zt * (ZL + j*Zt*tan(beta*l))
//       / (Zt + j*ZL*tan(beta*l))
// ------------------------------------------------------

f = linspace(0.5e9, 2.5e9, 1000);

lambda = c ./ f;
beta = 2 * %pi ./ lambda;

// Electrical length of transformer
theta = beta * l;

// Input impedance
```

```
Zin = Zt .* (ZL + %i*Zt.*tan(theta)) ./ ...
      (Zt + %i*ZL.*tan(theta));

// Magnitude of input impedance
Zin_mag = abs(Zin);

// ------------------------------------------------
// GRAPH
// ------------------------------------------------

scf(1);
plot(f/1e9, Zin_mag, "b", "LineWidth", 2);

xgrid();
xlabel("Frequency (GHz)");
ylabel("|Z_{in}| (Ohm)");
title("Input Impedance of Quarter-Wave Transformer");

a = gca();
a.data_bounds = [0.5, 30; 2.5, 90];

// Mark design frequency
plot([1.5 1.5], [30 90], "r--", "LineWidth", 1.5);
plot(1.5, Z0, "ko");

legend(["Input impedance"; ...
        "Design frequency"; ...
        "75 Ohm matched impedance"], "in_upper_right");
```

```
disp("==============================================");
disp("Graph plotted successfully.");
disp("At 1.5 GHz, Zin is approximately 75 Ohm.");
disp("==============================================");
```

OUTPUT: